**teffects psmatch — Propensity-score matching**

**Description**

*teffects psmatch* estimates the average treatment effect (ATE) and average treatment effect on the treated (ATET) from observational data by propensity-score matching (PSM). PSM estimators impute the missing potential outcome for each subject by using an average of the outcomes of similar subjects that receive the other treatment level. Similarity between subjects is based on estimated treatment probabilities, known as propensity scores. The treatment effect is computed by taking the average of the difference between the observed and potential outcomes for each subject. *teffects psmatch* accepts a continuous, binary, count, fractional, or nonnegative outcome.

See [TE] *teffects intro* or [TE] *teffects intro advanced* for more information about estimating treatment effects from observational data.

**Quick start**

ATE of *treat* on *y* estimated by PSM using a logistic model for *treat* on *x* and indicators for levels of categorical variable *a*

```
teffects psmatch (y) (treat x i.a)
```

As above, but estimate the ATET

```
teffects psmatch (y) (treat x i.a), atet
```

ATE of *treat* using a heteroskedastic probit model for treatment

```
teffects psmatch (y) (treat x i.a, hetprobit(x i.a))
```

With 4 matches per observation

```
teffects psmatch (y) (treat x i.a), nneighbor(4)
```

**Menu**

Statistics > Treatment effects > Continuous outcomes > Propensity-score matching

Statistics > Treatment effects > Binary outcomes > Propensity-score matching

Statistics > Treatment effects > Count outcomes > Propensity-score matching

Statistics > Treatment effects > Fractional outcomes > Propensity-score matching

Statistics > Treatment effects > Nonnegative outcomes > Propensity-score matching
Syntax

```
teffects psmatch (ovar) (tvar tmvarlist [, tmodel]) [if] [in] [weight]
    [, stat options]
```

*ovar* is a binary, count, continuous, fractional, or nonnegative outcome of interest.
*tvar* must contain integer values representing the treatment levels.
*tmvarlist* specifies the variables that predict treatment assignment in the treatment model. Only two treatment levels are allowed.

<table>
<thead>
<tr>
<th><em>tmodel</em></th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Model</strong></td>
<td></td>
</tr>
<tr>
<td>logit</td>
<td>logistic treatment model; the default</td>
</tr>
<tr>
<td>probit</td>
<td>probit treatment model</td>
</tr>
<tr>
<td>hetprobit(<em>varlist</em>)</td>
<td>heteroskedastic probit treatment model</td>
</tr>
</tbody>
</table>

*tmodel* specifies the model for the treatment variable.

<table>
<thead>
<tr>
<th><em>stat</em></th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Stat</strong></td>
<td></td>
</tr>
<tr>
<td>ate</td>
<td>estimate average treatment effect in population; the default</td>
</tr>
<tr>
<td>atet</td>
<td>estimate average treatment effect on the treated</td>
</tr>
</tbody>
</table>
options Description

Model

nnneighbor(#) specify number of matches per observation; default is nnneighbor(1)

SE/Robust

vce(vcetype) vcetype may be

vce(robust [ , nn(#)]); use robust Abadie–Imbens standard errors with # matches

vce(iid); use independent and identically distributed Abadie–Imbens standard errors

Reporting

level(#) set confidence level; default is level(95)

display_options control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling

Advanced

caliper(#) specify the maximum distance for which two observations are potential neighbors

pstolerance(#) set tolerance for overlap assumption

osample(newvar) newvar identifies observations that violate the overlap assumption

control(# | label) specify the level of tvar that is the control

tlevel(# | label) specify the level of tvar that is the treatment

generate(stub) generate variables containing the observation numbers of the nearest neighbors

coefflegend display legend instead of statistics

tmvarlist may contain factor variables; see [U] 11.4.3 Factor variables.
by and statsby are allowed; see [U] 11.1.10 Prefix commands.
fweights are allowed; see [U] 11.1.6 weight.
coefflegend does not appear in the dialog box.
See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Options

Model

nnneighbor(#) specifies the number of matches per observation. The default is nnneighbor(1). Each individual is matched with at least the specified number of individuals from the other treatment level. nnneighbor() must specify an integer greater than or equal to 1 but no larger than the number of observations in the smallest group.

Stat

stat is one of two statistics: ate or atet. ate is the default.

ate specifies that the average treatment effect be estimated.

atet specifies that the average treatment effect on the treated be estimated.
**teffects psmatch — Propensity-score matching**

`vce(vcetype)` specifies the standard errors that are reported. By default, `teffects psmatch` uses two matches in estimating the robust standard errors.

`vce(robust [ , nn(#) ])` specifies that robust standard errors be reported and that the requested number of matches be used optionally.

`vce(iid)` specifies that standard errors for independent and identically distributed data be reported.

The standard derivative-based standard-error estimators cannot be used by `teffects psmatch`, because these matching estimators are not differentiable. The implemented methods were derived by Abadie and Imbens (2006, 2011, 2012); see *Methods and formulas*.

As discussed in Abadie and Imbens (2008), bootstrap estimators do not provide reliable standard errors for the estimator implemented by `teffects psmatch`.

**Reporting**

`level(#)`; see [R] Estimation options.

`display_options`: `noci`, `nopvalues`, `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `alldbalevels`, `nofvlabel`, `fvwrap(#)`, `fvwrapon(style)`, `cformat(%,fmt)`, `pformat(%,fmt)`, `sformat(%,fmt)`, and `nolstretch`; see [R] Estimation options.

**Advanced**

`caliper(#)` specifies the maximum distance at which two observations are a potential match. By default, all observations are potential matches regardless of how dissimilar they are.

In `teffects psmatch`, the distance is measured by the estimated propensity score. If an observation has no matches, `teffects psmatch` exits with an error.

`pstolerance(#)` specifies the tolerance used to check the overlap assumption. The default value is `pstolerance(1e-5)`. `teffects` will exit with an error if an observation has an estimated propensity score smaller than that specified by `pstolerance()`.

`osample(newvar)` specifies that indicator variable `newvar` be created to identify observations that violate the overlap assumption. Two checks are made to verify the assumption. The first ensures that the propensity scores are greater than `pstolerance(#)` and less than `1 - pstolerance(#)`. The second ensures that each observation has at least `nneighbor(#)` matches in the opposite treatment group within the distance specified by `caliper(#)`.

The `vce(robust, nn(#))` option also requires at least `#` matches in the same treatment group within the distance specified by `caliper(#)`.

The average treatment effect on the treated, option `atet`, using `vce(iid)` requires only `nneighbor(#)` control group matches for the treated group.

`control(# | label)` specifies the level of `tvar` that is the control. The default is the first treatment level. You may specify the numeric level `#` (a nonnegative integer) or the label associated with the numeric level. `control()` and `tlevel()` may not specify the same treatment level.

`tlevel(# | label)` specifies the level of `tvar` that is the treatment for the statistic `atet`. The default is the second treatment level. You may specify the numeric level `#` (a nonnegative integer) or the label associated with the numeric level. `tlevel()` may only be specified with statistic `atet`. `tlevel()` and `control()` may not specify the same treatment level.

`generate(stub)` specifies that the observation numbers of the nearest neighbors be stored in the new variables `stub1`, `stub2`, .... This option is required if you wish to perform postestimation based
on the matching results. The number of variables generated may be more than \texttt{nneighbor(#)} because of tied distances. These variables may not already exist.

The following option is available with \texttt{teffects psmatch} but is not shown in the dialog box: \texttt{coeflegend}; see \texttt{[R] Estimation options}.

## Remarks and examples

Propensity-score matching uses an average of the outcomes of similar subjects who get the other treatment level to impute the missing potential outcome for each subject. The ATE is computed by taking the average of the difference between the observed and potential outcomes for each subject. \texttt{teffects psmatch} determines how near subjects are to each other by using estimated treatment probabilities, known as propensity scores. This type of matching is known as propensity-score matching (PSM).

PSM does not need bias correction, because PSM matches on a single continuous covariate. In contrast, the nearest-neighbor matching estimator implemented in \texttt{teffects nnmatch} uses a bias-correction term when matching on more than one continuous covariate. In effect, the PSM estimator parameterizes the bias-correction term in the treatment probability model. See \texttt{[TE] teffects intro} or \texttt{[TE] teffects intro advanced} for more information about this estimator.

We will illustrate the use of \texttt{teffects psmatch} by using data from a study of the effect of a mother’s smoking status during pregnancy (\texttt{mbsmoke}) on infant birthweight (\texttt{bweight}) as reported by Cattaneo (2010). This dataset also contains information about each mother’s age (\texttt{mage}), education level (\texttt{medu}), marital status (\texttt{mmarried}), whether the first prenatal exam occurred in the first trimester (\texttt{prenatal1}), whether this baby was the mother’s first birth (\texttt{fbaby}), and the father’s age (\texttt{fage}).

### Example 1: Estimating the ATE

We begin by using \texttt{teffects psmatch} to estimate the ATE of \texttt{mbsmoke} on \texttt{bweight}. We use a logistic model (the default) to predict each subject’s propensity score, using covariates \texttt{mage}, \texttt{medu}, \texttt{mmarried}, and \texttt{fbaby}. Because the performance of PSM hinges upon how well we can predict the propensity scores, we will use factor-variable notation to include both linear and quadratic terms for \texttt{mage}, the only continuous variable in our model:

```stata
. use https://www.stata-press.com/data/r16/cattaneo2
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu)
```

<table>
<thead>
<tr>
<th>Treatment-effects estimation</th>
<th>Number of obs = 4,642</th>
</tr>
</thead>
<tbody>
<tr>
<td>Estimator</td>
<td>propensity-score matching</td>
</tr>
<tr>
<td>Outcome model</td>
<td>matching</td>
</tr>
<tr>
<td>Treatment model: logit</td>
<td>max = 74</td>
</tr>
</tbody>
</table>

| bweight | AI Robust | z  | P>|z| | [95% Conf. Interval] |
|---------|-----------|----|------|---------------------|
| ATE     |           |    |      |                     |
| mbsmoke (smoker vs nonsmoker) | -210.9683 | 32.021 | -6.59 | 0.000 | -273.7284 | -148.2083 |

The average birthweight if all mothers were to smoke would be 211 grams less than the average that would occur if none of the mothers had smoked.
By default, teffects psmatch estimates the ATE by matching each subject to a single subject with the opposite treatment whose propensity score is closest. Sometimes, however, we may want to ensure that matching occurs only when the propensity scores of a subject and a match differ by less than a specified amount. To do that, we use the caliper() option. If a match within the distance specified in the caliper() option cannot be found, teffects psmatch exits.

Example 2: Specifying the caliper

Here we reconsider the previous example, first specifying that we only want to consider a pair of observations a match if the absolute difference in the propensity scores is less than 0.03:

```
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu),
> caliper(0.03)
```

The error arose because there is not a smoking mother whose propensity score is within 0.03 of the propensity score of the nonsmoking mother in observation 4504. If we instead raise the caliper to 0.10, we have matches for all subjects and therefore obtain the same results as in example 1:

```
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu),
> caliper(0.1)
```

Technical note

Example 2 highlights that estimating the ATE requires finding matches for both the treated and control subjects. In contrast, estimating the ATET only requires finding matches for the treated subjects. Because subject 4504 is a control subject, we can estimate the ATET using caliper(0.03). We must also specify vce(iid) because the default robust standard errors for the estimated ATET require viable matches for both treated subjects and control subjects. (This requirement comes from the nonparametric method derived by Abadie and Imbens [2012].)
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu), > atet vce(iid) caliper(0.03)

| bweight          | Coef.  | Std. Err. | z    | P>|z|    | [95% Conf. Interval] |
|------------------|--------|-----------|------|--------|---------------------|
| ATET mbsmoke     | -236.7848 | 26.11698  | -9.07 | 0.000  | -287.9731 -185.5964 |
| smoker vs nonsmoker |        |           |      |        |                     |

In the previous examples, each subject was matched to at least one other subject, which is the default behavior for `teffects psmatch`. However, we can request that `teffects psmatch` match each subject to multiple subjects with the opposite treatment level by specifying the `nneighbor()` option. Matching on more distant neighbors can reduce the variance of the estimator at a cost of an increase in bias.

> Example 3

Now we request that `teffects psmatch` match a mother to four mothers in the opposite treatment group:

. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu), > nneighbor(4)

| bweight          | Coef.  | Std. Err. | z    | P>|z|    | [95% Conf. Interval] |
|------------------|--------|-----------|------|--------|---------------------|
| ATE mbsmoke      | -224.006 | 29.88627  | -7.50 | 0.000  | -282.582 -165.43   |
| smoker vs nonsmoker |       |           |      |        |                     |

These results are similar to those reported in example 1.

**Video example**

Treatment effects in Stata: Propensity-score matching
Stored results

teffects psmatch stores the following in e():

Scalars
- \( e(N) \): number of observations
- \( e(n_j) \): number of observations for treatment level \( j \)
- \( e(k\_levels) \): number of levels in treatment variable
- \( e(caliper) \): maximum distance between matches
- \( e(treated) \): level of treatment variable defined as treated
- \( e(control) \): level of treatment variable defined as control
- \( e(k\_nnneighbor) \): requested number of matches
- \( e(k\_nnmin) \): minimum number of matches
- \( e(k\_nnmax) \): maximum number of matches
- \( e(k\_robust) \): matches for robust VCE

Macros
- \( e(cmd) \): teffects
- \( e(cmdline) \): command as typed
- \( e(depvar) \): name of outcome variable
- \( e(tvar) \): name of treatment variable
- \( e(subcmd) \): psmatch
- \( e(tmodel) \): logit, probit, or hetprobit
- \( e(stat) \): statistic estimated, ate or atet
- \( e(wtype) \): weight type
- \( e(wexp) \): weight expression
- \( e(title) \): title in estimation output
- \( e(tlevels) \): levels of treatment variable
- \( e(psvarlist) \): variables in propensity-score model
- \( e(hvarlist) \): variables for variance, only if hetprobit
- \( e(vce) \): vcetype specified in vce()
- \( e(vcetype) \): title used to label Std. Err.
- \( e(datasignature) \): the checksum
- \( e(datasignaturevars) \): variables used in calculation of checksum
- \( e(properties) \): \( b \ V \)
- \( e(estat\_cmd) \): program used to implement estat
- \( e(predict) \): program used to implement predict
- \( e(marginsnotok) \): predictions disallowed by margins
- \( e(asbalanced) \): factor variables fvset as asbalanced
- \( e(asobserved) \): factor variables fvset as asobserved

Matrices
- \( e(b) \): coefficient vector
- \( e(V) \): variance–covariance matrix of the estimators
- \( e(bps) \): coefficient vector from propensity-score model
- \( e(Vps) \): variance–covariance matrix of the estimators from propensity-score model

Functions
- \( e(sample) \): marks estimation sample

Methods and formulas

The methods and formulas used by teffects psmatch are documented in the Methods and formulas of [TE] teffects nnmatch.

References


Also see

[TE] teffects postestimation — Postestimation tools for teffects

[TE] teffects — Treatment-effects estimation for observational data

[TE] teffects nnmatch — Nearest-neighbor matching

[U] 20 Estimation and postestimation commands